MOYENNES et INTEGRATIONS NUMERIQUES

R. Coquidé (09/05/2018)

OUTILS GÉNÉRAUX

E P nat MD de BOITAOUTILS

```
I = : ] :. ]
```

Fonction Identité

Ex :
```
   I 5 2 8
5 2 8
```

```
pc =: ]%+/
```

Calcul des coefficients de pondération (Somme = 1) en fonction des poids (Somme non nulle).

Ex :
```
   pc 5 7 3 9 22 4
0.1 0.14 0.06 0.18 0.44 0.08
```

DÉFINITIONS MATHÉMATIQUES

Soit f une fonction strictement monotone continue (*donc inversible*) dans un intervalle fini $I \subset \mathbb{R}$.

Soit Y= $y_0\ y_1\ y_2 \dots y_{n-1}$ une suite finie de n nombres situés dans *I*.

La f-moyenne de Y est $\boxed{m = f^{-1}\left(\frac{1}{n}\sum_{i=0}^{n-1} f(y_i)\right)}$

Soit C=$c_0\ c_1\ c_2 \dots c_{n-1}$ une suite de coefficients de pondération (Somme=1).

La f-moyenne de Y pondérée par les coefficients de pondération C est :

$$\boxed{m = f^{-1}\left(\sum_{i=0}^{n-1} c_i . f(y_i)\right)}$$

La f-moyenne glissante de pas k est :

$$\boxed{m = m_0\ m_1\ m_2 ... m_{n-k} \text{ tels que } m_i = f^{-1}\left(\frac{1}{k}\sum_{j=0}^{k-1} f(y_{i+j})\right)}$$

La f-moyenne glissante pondérée de pas k et coefficients de pondération C=$c_0 \dots c_{k-1}$ est :

$$\boxed{m = m_0\ m_1\ m_2 ... m_{n-k} \text{ tels que } m_i = f^{-1}\left(\sum_{j=0}^{k-1} c_j . f(y_{i+j})\right)}$$

PROGRAMMATION DE QUELQUES MOYENNES

```
Moy = : (+/%#)&.:
```

Pro-adverbe permettant le calcul d'une f-moyenne sous la forme :

`m =. f Moy Y`   NB. Y est une suite finie de nombres réels

Pro-verbes calculant les moyennes les plus classiques :

```
ma =:  I Moy      NB. Moyenne arithmétique
mg =: ^. Moy      NB. Moyenne géométrique
mh =:  % Moy      NB. Moyenne harmonique
mq =: *: Moy      NB. Moyenne quadratique
```

Ex :

```
V =. 5 7 4 9 2 4 5 3 6 1 8 9 6 7 4 9 2
(mh ; mg ; ma ; mq)  V
```

3.72749	4.61794	5.35294	5.90613

```
MoyP =: ([:+/[@*)&.:
```

Pro-adverbe permettant le calcul des moyennes pondérées

Pro-verbes calculant les moyennes pondérées les plus classiques :

```
maP =:  I MoyP      NB. Arithmétique Pondérée
mgP =: ^. MoyP      NB. Géométrique Pondérée
mhP =:  % MoyP      NB. Harmonique Pondérée
mqP =: *: MoyP      NB. Quadratique Pondérée
```

Ex :

```
V =. 5 7 8 3 12 4 14 7 2 1
C =. pc p =. 12 3 6 9 11 15 7 5 3 5
C  (maP ; mgP ; mhP ; mqP) V
```

6.47	3.97e_17	0.0238	2.6

```
MoyG =: Moy\
```

Pro-adverbe permettant le calcul de moyennes glissantes

Pro-verbes calculant les moyennes glissantes les plus classiques :

```
maG  =:  I MoyG      NB.  Moyenne arithmétique glissante
mgG  =: ^. MoyG      NB.  Moyenne géométrique glissante
mhG  =:  % MoyG      NB.  Moyenne harmonique  glissante
mqG  =: *: MoyG      NB.  Moyenne quadratique  glissante
```

Ex : Moyennes glissantes de pas 3

```
V =. 5 7 4 9 2 4 5 3 6 1 8 9 6
3  maG  V              NB. arithmétique
5.33 6.67 5 5 3.67 4 4.67 3.33 5 6 7.67
```

```
    3  mgG  V           NB. géometrique
 5.19 6.32 4.16 4.16 3.42 3.91 4.48 2.62 3.63 4.16 7.56
    3  mhG  V           NB. harmonique
 5.06 5.95 3.48 3.48 3.16 3.83 4.29 2 2.32 2.43 7.45
    3  mqG  V           NB. quadratique
 5.48 6.98 5.8 5.8 3.87 4.08 4.83 3.92 5.8 6.98 7.77
```

```
MoyPG =: 1 : 'u^:_1@([:+/[*[:|:[:>#@[<\u@])'
```

Pro-adverbe pour le calcul des moyennes pondérées glissantes

Pro-verbes calculant des moyennes pondérées glissantes les plus classiques

```
maPG =: I  MoyPG   NB. Arithmétique Pondérée Glissante
mgPG =: ^. MoyPG   NB. Géométrique Pondérée Glissante
mhPG =: %  MoyPG   NB. Harmonique Pondérée Glissante
mqPG =: *: MoyPG   NB. Quadratique Pondérée Glissante
```

Ex :
```
Y =. 6 9 8 2 8 12 4 7 9   NB. Suite de nombres
X =. 1r2 1r6 1r3          NB. Coef de pondération (somme 1)
8j2":   X (maPG,.mgPG,.mhPG,.mqPG) Y
```

7.17	7.07	6.97	7.27
6.50	5.35	4.11	7.25
7.00	6.35	5.33	7.35
6.33	4.58	3.35	7.79
7.33	6.79	6.26	7.83
9.00	8.35	7.64	9.54
6.17	5.75	5.38	6.57

MOYENNES HYBRIDES

```
mag =: {.@((mg,ma)^:_) NB. Arithmético-Géométrique de 2 nombres
mah =: {.@((mh,ma)^:_) NB. Arithmético-Harmonique de 2 nombres
mhg =: {.@((mh,mg)^:_) NB. Harmonico-Géométrique de 2 nombres

magI =: {.@(E@(mg,ma)^:_) NB. Affiche les Itérations de mag
mahI =: {.@(E@(mh,ma)^:_) NB. Affiche les Itérations de mah
mhgI =: {.@(E@(mh,mg)^:_) NB. Affiche les Itérations de mhg
```

Ex : moyenne arithmético-géométrique :

```
    mag 5.2 8.7
 6.83757597564565
    magI 5.2 8.7
6.72606868832009 6.95
6.83711762249449 6.83803434416005
6.83757596796403 6.83757598332727
6.83757597564565 6.83757597564565
6.83757597564565 6.83757597564565
6.83757597564565
```

Pro-verbe calculant la moyenne d'ordre k (k réel non nul) :

```
moyk=:(([:+/^~)%[:#])^[:%[
```

Ex :
```
        Y =. 6 9 8 2 8 12 4 7 9
        1r2   moyk   Y
    6.89378147522855
        3   moyk   Y
    8.13262706996978
```

MOYENNE ET INTEGRALE D'UNE FONCTION SUR UN INTERVALLE [a , b]

UTILITAIRES GENERAUX

Découpage de l'intervalle y en x sous-intervalles égaux (pro-verbe)

```
coupe =: {.@]+i.@>:@[*({:-{.)@]%[
```

Ex :
```
        10 coupe 2 4
    2  2.2  2.4  2.6  2.8  3  3.2  3.4  3.6  3.8  4
```

Transformation affine de la suite croissante x située dans [_1,1] vers l'intervalle y

```
traf =: 4 : '0.5*(+/y)-x*(-/y)'"0 _
```

Ex :
```
        _0.8  _0.6  0.2  0.4  0.7  0.9 traf  5  9
    5.4  5.8  7.4  7.8  8.4  8.8
```

PRO-VERBE étendue d'une suite de réels

```
etendue =: >./ - <./
```

Ex :
```
        etendue Y=.6 9 8 2 8 12 4 7 9
    10
```

PRO-VERBE qui transforme une suite croissante de points dans un intervalle en une suite de sous intervalles (1 par ligne)

```
mat =: >@(2&(<\))
```

Ex :
```
        mat 1 2 3 4
    1 2
    2 3
    3 4
```

PRO-VERBE partition de l'intervalle y en sous-intervalles

```
part =: mat@(] : ([coupe]))
```

Ex1 : `part 2 4 6 8 10` NB. Sous-intervalles pré-définis
```
    2  4
    4  6
    6  8
    8  10
```

Ex2 : `5 part 3 10` NB. 5 sous-intervalles égaux
```
    3    4.4
    4.4  5.8
    5.8  7.2
    7.2  8.6
    8.6  10
```

METHODE DE SIMPSON

Approximation de la fonction par des portions de paraboles passant par 3 points

```
ISIMPS =: 2 : '(+/(1,((n1-1)$4 2),1)*u({.y)+h*i.1+n1)*(h=.-(-/y)%n1=.2*n)%3'
```

Pro-conjonction : Intégrale selon la méthode de SIMPSON

Utilisation : `R=.  (f ISIMPS  N)  a,b`

R(résultat), f(fonction à intégrer : pro-verbe), N(nb. de portions de paraboles)
a,b(intervalle)

Ex : Intégrale $\int_1^5 t.e^{-t} dt$

```
        f =: 3 : 'y*^-y'
        (f ISIMPS 50) 1 5
    0.695
```

```
MSIMPS =: 2 : '((u ISIMPS n)y)% etendue y'
```

Pro-conjonction : Moyenne selon la méthode de simpson

Ex : Moyenne de f sur l'intervalle 1 5

```
        (f MSIMPS 50) 1 5
0.174
```

METHODE DE NEWTON-COTES (NC)

(Cette méthode utilise un polynôme dont les racines sont réelles, équidistantes, situées dans l'intervalle d'intégration).

OUTILS POUR METHODE DE NEWTON-COTE (NC)

Racines dans [_1,1] pour NC :

```
rnc =: <:@(]%~2:*i.@>:)
```

Matrice de VANDERMONDE pour NC

```
vdmNC =: [:|:rnc^/i.@>:
```

Seconds membres pour NC :

```
smNC =: (>:$ 1 0"_)%>:@i.@>:
```

Coef. de pondération pour NC :

cNC =: smNC %. vdmNC

Utilisation: vdmNC n ; smNC n ; cNC n ; n entier de 1 à 7

```
fnc =: cNC@[maP]
```
NB. PRO-VERBE

Abscisses dans [a,b] pour NC :

```
aNC =: rnc@[traf]
```

Utilisation: n fnc y0 y1 ...yn ; n aNC a,b

APPROXIMATIONS : MOYENNE ET INTEGRALE DE NEWTON-COTES

```
MNC =: 2 : '+/(cNC n)*(u (n aNC y))'
```

Moyenne par la méthode de Newton-Cotes

Utilisation : `R =. ( f  MNC  N ) a,b`

Où f : fonction (verbe) dont on calcule la moyenne sur l'intervalle réel [a,b]

N : degré du polynôme d'interpolation (suggestion : $1 \leq N \leq 7$)

Ex :

```
f =: 3 : 'y*^-y'
( f  MNC 7) 1 5
0.173825
```

NB. $f(t) = t.e^{-t}$

NB. Moyenne sur [1,5]

```
INC =: 2 : '((u MNC n)y)*({:-{.)y'
```

Intégrale par la méthode de NEWTON-COTES

Utilisation : `R  =. ( f INC N ) a,b`

Ex :

```
f =: 3 : 'y*^-y'
( f INC 7 ) 1 5
0.695301
```

NB. $f(t) = t.e^{-t}$

NB. Intégrale sur [1,5]

```
INCC  =:  2 : ('[:',MD,'+/(u INC n)"1 x part y')
```

Intégrale Newton-Cotes Composite

Utilisation : `R =. N1 ( f INCC N ) a,b`

Où f : fonction (verbe) à intégrer sur l'intervalle fini [a,b] lui-même divisé en N1 sous-intervalles de longueurs égales ; sur chacun d'eux, on utilise un polynôme d'interpolation de degré N ($1 \leq N \leq 7$)

Ex : `f =: 3 : 'y*^-y'` NB. $f(t) = t.e^{-t}$

```
   5 ( f INCC 7 ) 1 5
0.69533120020854
   10 ( f INCC 7 ) 1 5
0.695331200347815
   20 ( f INCC 7 ) 1 5
0.695331200348371
```

NB. Valeur exacte : $2e^{-1} - 6e^{-5} =$

NB. 0.695331200348372…

```
MNCC =: 2 : ('[:',MD,'(x (u INCC n) y) % etendue y'
```

Moyenne Newton-Cotes Composite (Mêmes paramètres que INCC)

Ex :

```
   20 ( f MNCC 7 ) 1 5
0.173832800087093
```

METHODE DE GAUSS-LEGENDRE (GL)
(Cette méthode utilise des polynômes de Legendre)

OUTILS POUR METHODES DE GAUSS-LEGENDRE (GL)

Coef. des Polynômes de Legendre :
```
    plg =: 3 : 0
r=.p=.1[ q=.i.k=.0
while. (y>:k=.>:k) do.
p=.r[q=.p[r=.(((_1+2*k)*p,0)-(k-1)*0 0,q)%k
end.  r
)
```

Racines des polynômes de Legendre :
```
    rpl =: [:/:~>@(1:{p.@|.@plg)
```
Utilisation: plg n ; rpl n ; (n entier de 1 à 7)

Matrice de VANDERMONDE pour GL :
```
    vdmGL =: [:|:rpl^/i.
```

Seconds membres pour GL :
```
    smGL =: (]$1 0"_)%>:@i.
```

Coefficients de pondération pour GL
```
    cGL =: smGL%.vdmGL
```
Utilisation: vdmGL n ; smGL n ; cGL n ; n entier de 1 7

Pro-verbe :
```
    fgl =: cGL@[maP]
```

Abscisses dans[a,b] pour méthode de Gauss-Legendre
```
    aGL =: rpl@[traf]
```

APPROXIMATIONS : MOYENNE ET INTEGRALE DE GAUSS-LEGENDRE

```
MGL =: 2 : '(n&fgl) u (n aGL y)'
```
Moyenne Gauss-Legendre

Utilisation : `R =. ( f  MGL  N ) a,b`

Où f : fonction (verbe) dont on calcule la moyenne sur l'intervalle réel [a,b]

N : degré du polynôme d'interpolation (suggestion : $1 \leq N \leq 7$)

Ex : `f =: 3 : 'y*^-y'` NB. $f(t) = t.e^{-t}$

`( f  MGL 5 ) 1 5` NB. Moyenne sur [1,5]
```
    0.17383
```

```
IGL =: 2 : '(etendue y)*(u MGL n) y'
```
Intégrale Gauss-Legendre

Utilisation : `R  =. ( f INC N ) a,b`

Ex : `( f IGL 7 ) 1 5` NB. Intégrale sur [1,5]

```
0.695301
```

```
IGLC =: 2 : ('[:',MD,'+/(u IGL n)"1 (x part y)')
```

Intégrale Gauss-Legendre Composite

Utilisation : `R =. N1 ( f IGLC N ) a,b`

Où f : fonction (verbe) à intégrer sur l'intervalle fini [a,b] lui-même divisé en N1 sous-intervalles de longueurs égales ; sur chacun d'eux, on utilise un polynôme d'interpolation de degré N ($1 \leq N \leq 7$)

Ex : `f =: 3 : 'y*^-y'` NB. $f(t) = t.e^{-t}$

```
   5 ( f IGLC 7 ) 1 5
0.695331200348372
```

```
MGLC =: 2 : ('[:',MD,'(x(u IGLC n)y)%(etendue y)')
```

Moyenne Gauss-Legendre Composite (Mêmes paramètres que IGLC)

Ex :

```
   10 ( f MGLC 7 ) 1 5
0.173832800087093
```

METHODE DE TCHEBYCHEV (TC)

(Cette méthode utilise des polynômes de TCHEBYCHEV)

OUTILS POUR METHODE DE TCHEBYCHEV (TC)

Abscisses de TCHEBYCHEV dans l'intervalle [_1,1] :

```
    atc =: 3 : 0
s=. |.(y $(k=.0),c=.1)*y%2+i.y
while. (0<:y+k=.<:k) do. c=.c,(c+/ .*k{.s)%k end.
/:~>1{p.|.c
)
```

Abscisses dans [a,b] pour méthode de TChebychev

```
    aTC =: atc@[traf]
```

APPROXIMATIONS : MOYENNE ET INTEGRALE DE TCHEBYCHEV

```
MTC =: 2 : 'ma (u(n&aTC y))'
```

Moyenne TChebychev

Utilisation : `R =. ( f  MTC  N ) a,b`

Où f : fonction (verbe) dont on calcule la moyenne sur l'intervalle réel [a,b]

N : degré du polynôme d'interpolation (suggestion : $1 \leq N \leq 7$)

Ex : `f =: 3 : 'y*^-y'` NB. $f(t) = t.e^{-t}$

```
    ( f  MGL 3 ) 1 5
0.1742
```

NB. Moyenne sur [1,5]

```
ITC =: 2 : '(etendue y)*(u MTC n y)'
```

Intégrale TChebychev

Utilisation : `R  =. ( f INC N ) a,b`

Ex : `( f ITC 4 ) 1 5` NB. Intégrale sur [1,5]

```
0.6964
```

```
ITCC =: 2 : ('[:',MD,'+/((u ITC n)"1)(x part y)')
```

Intégrale TChebychev Composite

Utilisation : `R =. N1 ( f ITCC N ) a,b`

Où f : fonction (verbe) à intégrer sur l'intervalle fini [a,b] lui-même divisé en N1 sous-intervalles de longueurs égales ; sur chacun d'eux, on utilise un polynôme d'interpolation de degré N ($1 \leq N \leq 7$)

Ex : `f =: 3 : 'y*^-y'` NB. $f(t) = t.e^{-t}$

```
    5 ( f ITCC 7 ) 1 5
0.69533120044848l
```

```
MTCC =: 2 : ('[:',MD,'(x(u ITCC n)y)%(etendue y)')
```

Moyenne TChebychev Composite (Mêmes paramètres que ITCC)

Ex : `10 ( f MGLC 7 ) 1 5`

```
0.173832800087093
```

UNE METHODE PRATIQUE

INTEGRATION PAR 3 METHODES COMPOSITES

```
INTC =: 2 : ('[:',MD,',.(x(u INCC n)y),(x(u IGLC n)y),
(x(u ITCC n)y)')
```

Intégration composite : méthode pratique (comparaison des 3 méthodes d'intégration)
On utilise en parallèle les trois méthodes de NEWTON-COTES, GAUSS-LEGENDRE, Et TCHEBYCHEV. Elles ont en commun d'être composites (l'intervalle d'intégration est partitionné en sous-intervalles de mêmes longueurs). Elles diffèrent par le choix du type de polynôme d'interpolation dans chaque sous-intervalle, et des coefficients de pondération.

Utilisation : `R =. N1 ( f INTC N ) a,b`

Où f : fonction (verbe) à intégrer sur l'intervalle fini [a,b] lui-même divisé en N1 sous-intervalles de longueurs égales ; sur chacun d'eux, on utilise un polynôme d'interpolation de degré N ($1 \le N \le 7$) différent pour chaque méthode d'intégration (NEWTON-COTES , GAUSS-LEGENDRE, TCHEBYCHEV).

Ex : On veut intégrer la fonction suivante entre 2 et 5 :

$$fi(t) = \frac{2te^{-3.2t} + 3\sin(0.13t^{1.7})}{5.2 + 3.1\sqrt{5+t^2}}$$ (on ne connaît aucune méthode analytique)

```
fi =: 3 : '((2*y*^_3.2*y)+(3*1 o. 0.13*y^1.7))% (5.2+3.1*
%:5+*:y)'
```

```
        3  (fi INTC 1) 2 5          NB. 3 sous-intervalles, degré 1
0.394578864236555                   ← NEWTON-COTES
0.399493068483228                   ← GAUSS-LEGENDRE
0.399493068483228                   ← TCHEBYCHEV
        5  (fi INTC 3) 2 5          NB. 5 sous-intervalles, degré 3
0.397853796435875
0.397852829077372
0.397852287315627
        10 (fi INTC 5) 2 5          NB. 10 sous-intervalles, degré 5
0.397852830558561
0.397852830545346
0.397852830536872
        10 (fi INTC 7) 2 5          NB. 10 sous-intervalles, degré 7
0.397852830545342
0.397852830545346
0.397852830545349
        20 (fi INTC 7) 2 5          NB. 20 sous-intervalles, degré 7
0.397852830545347
0.397852830545346
0.397852830545346
        70 (fi INTC 7) 2 5          NB. 70 sous-intervalles, degré 7
0.397852830545346
0.397852830545346
0.397852830545346
```

Remarques : Les chiffres soulignés peuvent être considérés comme exacts.
On constate que la méthode de GAUSS-LEGENDRE est généralement la plus précise (pour les mêmes valeurs de N et N1).
Si les 3 méthodes donnent les k premières décimales identiques, on peut les considérer comme exactes avec une quasi certitude.
Pour N=1, les méthodes de GAUSS-LEGENDRE et TCHEBYCHEV donnent exactement le même résultat : c'est normal puisque les polynômes de LEGENDRE et TCHEBYCHEV de degré 1 sont identiques !
Au-delà du degré 7, les méthodes sont instables, donc peu fiables.
Si on augmente N1, nombre de sous-intervalles, l'erreur diminue jusqu'à un optimum puis augmente lentement mais indéfiniment... et tend vers l'infini avec N1 (!!!) : cela provient du fait que l'erreur systématique tend vers 0 quand N1 tend vers l'infini mais que l'erreur d'arrondi tend vers l'infini.

MOYENNE PONDEREE D'UNE FONCTION

```
MPF =: 1 : '(50&(((m@.0)*(m@.1))IGLC 7))%
(50&((m@.0)IGLC 7))'
```

Pro-adverbe pour le calcul de la moyenne d'une fonction pondérée par une fonction de poids strictement positive sur un intervalle fini [a,b]. On utilise la méthode GL.

Utilisation : `R=. fp`f MPF a,b`

Où : fp : fonction de poids strictement positive sur [a,b]

f : fonction dont on calcule la moyenne pondérée

[a,b] : intervalle de longueur finie

Ex :

```
mu =: 3 : 'y*1 o.3*y'
fp =: 3 : '^1r3**:y'
fp`mu MPF 2 7
5.61897108
```

NB. $mu(t) = t\sin(3t)$

NB. $fp(t) = e^{\frac{t^2}{3}}$ fonction de poids

NB. Moyenne pondérée sur [2,7]

FONCTION-DENSITE D'UNE FONCTION > 0 SUR UN INTERVALLE FINI

```
DENS =: 2 : ('[:',MD,'(u y)%(50(u IGLC n)x)')
```

Pro-conjonction pour le calcul d'une fonction-densité sur un intervalle fini [a,b]
On utilise la méthode de Gauss-Legendre.

Utilisation : `fdens =: (a,b)& (f DENS 7)`

Où fdens : fonction-densité cherchée sur l'intervalle [a,b]

f : fonction strictement positive sur cet intervalle

Ex :

```
psi =: 3 : '*: y+^y*2 o.y'
psidens =: (1 6) & ( psi DENS 7 )
12j8 ": ,. psidens 2 2.7 4.2 5.3
0.00035737
0.00046817
0.00112873
0.03524279
10 ( psidens IGLC 7) 1 6
0.999999999960734
30 ( psidens IGLC 7) 1 6
0.999999999999999
40 ( psidens IGLC 7) 1 6
1
```

NB. $psi(t) = (t + e^{t\cos(t)})^2$

NB. L'intégrale sur son intervalle
NB. de définition d'une fonction-
NB. densité est égale à 1

6 PRO-VERBES CALCULANT L'ECART-TYPE D'UNE SUITE DE NOMBRES

```
ect1 =: [:%:(([:ma*:)-[:*:ma)
ect2 =: %:@:(ma@:*:-*:@:ma)
ect3 =: %:@:ma@:*:@:(ma-])
ect4 =: [:%:[:ma[:*:ma-]
ect5 =: ma&.:*:@(ma-])
ect6 =: mq@(ma-])
```

```
Ex:        s=.5 7 3 9 8 1 12 4 9 15 3 7 2 10
           10j6 ": ,. (ect1,ect2,ect3,ect4,ect5,ect6) s
3.894659
3.894659
3.894659
3.894659
3.894659
3.894659
```